
11

ÎNTRERUPERI

11.1 Întreruperi interne şi externe

O întrerupere este un eveniment care duce la întreruperea temporară a execuţiei unui program, executând o subrutina numită serviciul întreruperii (Interrupt Service Routine – ISR), după care reia programul oprit din acelaşi punct, ca şi când nu s-ar fi întâmplat nimic. Întreruperile au fost realizate ca alternativă la examinarea repetată a perifericelor pentru detectarea unei operaţii de intrare/ieşire (polling). Astfel, în locul acestei interogări repetate şi costisitoare, perifericele generează semnale de întrerupere, apelând serviciul corespunzător.

Deşi momentul de declanşare este oarecare, semnalele de întrerupere nu sunt recunoscute de către procesor decât între doua apeluri de instrucţiuni. Astfel, dacă se execută o instrucţiune ce necesită multe cicluri maşina (cum este, spre exemplu, o înmulţire, care poate necesita până la 139 de cicluri procesor), iar un apel de întrerupere are loc în timpul acesta, el este pus în aşteptare până la terminarea instrucţiunii. Excepţie de la această regula o fac instrucţiunile cu repetare, cum sunt instrucţiunile de lucru cu şiruri (de exemplu rep movsb), care sunt întrerupte.

În cadrul programării procesoarelor din familia 8086, întreruperile se împart în două clase:

· întreruperi externe

· întreruperi interne

Întreruperile externe au loc în momentul generării unui semnal de către diferitele dispozitive spre procesor. Pe de altă parte, întreruperile interne sunt declanşate în două moduri: ca urmare a apelului unei instrucţiuni int sau ca urmare a unei excepţii generate de o condiţie de eroare (cum ar fi spre exemplu împărţirea la zero). Întreruperile interne generate prin apeluri int se mai numesc întreruperi software.

11.2 Modul de funcţionare a întreruperilor

Un serviciu oferit de o întrerupere nu este altceva decât un tip special de subprogram. Subprogramele implicite pot fi înlocuite cu rutine construite de către programator. Datorită faptului că aceste servicii nu sunt proceduri obişnuite, câteva măsuri trebuiesc luate în cadrul lor:

· salvarea tuturor regiştrilor la începutul rutinei

· apelul instrucţiunii sti pentru procesarea altor întreruperi în cadrul serviciului

· restaurarea regiştrilor la sfârşitul rutinei

· apelul iret ca ultimă instrucţiune

Salvarea şi restaurarea regiştrilor folosiţi este necesară datorită incertitudinii în ceea ce priveşte momentul în care au loc întreruperile externe. Astfel, întreruperile de acest tip pot avea loc în orice moment, întrerupând execuţia unui bloc de instrucţiuni, iar modificarea fără restaurare a regiştrilor folosiţi în acest bloc duce la rezultate catastrofale pentru program. Pe de altă parte, întreruperile software au un control mai mare relativ la momentul lor de execuţie şi se pot sustrage de la această regulă.

Dacă se doreşte ca şi alte întreruperi să fie capabile de activare în timpul execuţie serviciului, întrerupând astfel execuţia acestuia, se realizează apelul instrucţiunii sti, care are ca efect setarea indicatorului de condiţie IF (interrupt-enable flag). Dacă nu se realizează acest apel, iar dacă if nu este setat, întreruperile ce au loc în timpul rezolvării întreruperii curente, nu vor fi recunoscute de către procesor. Pentru resetarea (punerea pe zero) a acestui flag se foloseşte instrucţiunea cli. De remarcat, că aceste două instrucţiuni, nu au nici un efect asupra întreruperilor interne, software sau generate de o condiţie de eroare, ci numai asupra acceptării sau nu a întreruperilor externe. 

Familia de procesoare 8086 are doi pini de intrare de la dispozitivele generatoare de întreruperi:

· pentru întreruperi mascabile (Maskable Interrupts) – INTR

· pentru întreruperi nemascabile (Nonmaskable Interrupts) – NMI

Majoritatea dispozitivelor ce generează întreruperi folosesc linia INTR pentru semnalarea evenimentelor. Instrucţiunile cli şi sti afectează întreruperile ce sosesc pe acest pin: cli realizează o “mascare” a întreruperilor, prevenind recunoaşterea lor de către procesor, în timp ce sti permite intervenţia acestora în codul executat de către procesor. Nici una dintre aceste instrucţiuni nu are efect asupra liniei NMI, pe care circulă semnale importante, ce nu pot fi mascate, cum ar fi cele ce declanşează execuţia codul în cazul căderii tensiunii. În cadrul arhitecturii IMB PC originale, linia NMI era folosită pentru erorile de paritate a memorie RAM, cauzate de defectarea unuia sau mai multor biţi. Arhitecturile actuale oferă multe alte servicii noi pentru situaţii critice, făcând programarea serviciilor NMI complicată.

Deşi linia NMI pare de neblocat în sarcina sa de a trimite semnale de întrerupere către procesor, ea poate fi totuşi blocată prin diferite trucuri ce diferă de la o arhitectură la alta. Astfel, pentru IBM XT, se poate masca şi debloca această linie prin scrierea valorii 00h (pentru mascare) şi a valorii 080h (pentru deblocare) pe portul 0A0h.

Pentru întoarcerea în mod corect în programul întrerupt, se va folosi instrucţiunea iret în locul instrucţiunilor obişnuite de revenire în aplicaţie. Această instrucţiune, pe lângă sarcina de refacere a punctului de execuţie (refacerea din stivă a perechii CS:IP), realizează şi sarcini suplimentare specifice întreruperilor, cum ar fi refacerea flagurilor IF şi TF (trap flag). Pe lângă cele doua linii de transmitere a întreruperilor mai sus amintite, arhitectura IBM vine în ajutorul programatorilor cu un cip, numit Intel 8259 Programmable Interrupt Controller (PIC),  care serveşte până la opt dispozitive ce generează întreruperi. Arhitecturile mai noi au în dotare şi alte cipuri de tip PIC pentru mai multe dispozitive. Pentru arhitecturile IBM AT cu două cipuri PIC lista întreruperilor externe se prezintă astfel:

Tabel 11.1.


Nivelul PIC
Numarul Intreruperi
Dispozitiv

0
08h
Timer (ceasul software)

1
09h
Tastatura

2
0Ah
Slave 8259

3
0Bh
Portul serial secundar (COM2)

4
0Ch
Protul serial primar (COM1)

5
0Dh
Hard Disk (Fix)

6
0Eh
Floppy Disk

7
0Fh
Imprimanta pe port paralel

8
070h
Ceasul hardware

9
071h
Master 8259 Nivel 2

10
072h
-

11
073h
-

12
074h
-

13
075h
Coprocesorul numeric

14
076h
Hard Disk (Fix)

15
077h
-

NMI
02h
Paritate Memorie

*datorita faptului că întreruperile NMI sunt tot externe, aceasta linie 

a fost inclusa in tabel deşi nu este ataşată controller-ului 8259

Prima coloana a tabelului anterior, nivelul PIC, arată prioritatea serviciului: dacă două întreruperi au loc simultan, cotroller-ul 8259 dă prioritate celei cu acest număr mai mic.

A doua coloană, numită numărul întreruperii (tipul sau nivelul întreruperii), identifică ISR-ul (serviciul) ce rulează la declanşarea acesteia. Acesta este un index folosit de procesor pentru determinarea adresei rutinei, în timp ce nivelul PIC arată pinul din cadrul cipului 8259. Acest număr se regăseşte şi în cadrul întreruperilor software, el fiind folosit în instrucţiunile int ca argument.

Prin intermediul numărul instrucţiunii, procesorul află un pointer pe patru baiţi, numit vectorul întreruperii, care punctează rutina de tratare a întreruperii. În acest pointer, segmentul şi deplasamentul, sunt memorate la adresa mai mica, cu valori între 0000:0000 si 0000:30FF pentru arhitectura PC AT. În momentul în care un semnal de întrerupere este generat de unul de dispozitivele periferice, cipul 9259 activează linia INTR, aşteptând încuviinţarea procesorului pentru primirea întreruperii. La primirea încuviinţării, procesorul primeşte numărul întreruperii şi îl foloseşte pentru aflarea adresei rutinei de tratare pe care o apelează. 

În mod asemănător au loc operaţiile pentru apelul rutinei de tratare a unei întreruperi interne ca urmare a unei instrucţiuni int sau ca urmare a unei condiţii de eroare. Atât în cazul întreruperilor externe, cât şi în cel al celor interne, procesorul realizează următoarele operaţii când primeşte un număr de întrerupere:

· indicatorii de condiţie sunt salvaţi pe stivă

· indicatorii de condiţie IF si TF sunt setaţi pe zero

· regiştrii IP şi CS sunt salvaţi pe stivă

· vectorul întreruperii (adresa serviciului) este copiat în CS:IP

Odată cu iniţializarea regiştrilor CS:IP cu valoarea vectorului întreruperii, se trece la execuţia rutinei aflată la acea adresă. Programatorul poate modifica unul sau mai mulţi vectori de întreruperi şi, astfel, să îşi introducă propriile rutine de tratare. La terminarea unei rutine de tratare, se apelează instrucţiunea iret care realizează următoarele operaţii:

· regiştrii CS şi IP sunt restauraţi din stivă

· indicatorii de condiţie sunt refăcuţi, de asemenea, din stivă

Prima operaţie este operaţia obişnuită de refacere a adresei de revenire, pe care şi o instrucţiune obişnuită de retur o realizează. După care urmează refacerea indicatorilor de condiţie. Se observă că indicatorii de condiţie IF şi TF sunt resetaţi, la intrarea în rutină, după ce au fost salvate împreună cu celelalte flaguri. Astfel, pe parcursul execuţiei serviciului ele vor avea valoarea zero, nepermiţând întreruperea rutinei de alte întreruperi (indicatorul de condiţie IF) şi nici suport de întrerupere a instrucţiunilor pentru debuggere (indicatorul de condiţie TF – trap flag). Pentru permiterea întreruperii temporare a serviciului (setarea indicatorului de condiţie IF) se apelează instrucţiunea sti.

11.3 Tipuri de întreruperi 

O primă clasificare a întreruperilor este cea realizată în primul subcapitol: externe şi interne. Întreruperile se mai pot clasifica si după locul pe care îl ocupă rutinele de tratare implicite. Astfel, unele rutine de tratare se găsesc în BIOS, în timp ce altele sunt oferite de sistemul de operare. 

În cazul sistemului de operare MS-DOS, pe lângă shell-ul, ce interpretează comenzile, se găsesc doua straturi numite BDOS (Basic Disk Operating System) şi BIOS (Basic Input Output System). Pe disc acestea se gasesc în fişierele MSDOS.SYS si IOSYS.SYS. Aceste straturi cuprind rutine de tratare a întreruperilor, primul oferind rutine din partea sistemului de operare, în timp ce al doilea strat, BDOS, oferă o legătură spre întreruperile oferite de BIOS. Astfel, o alta clasificare a întreruperilor, pentru MS-DOS, este:

· întreruperi BIOS

· întreruperi DOS

11.3.1 Întreruperi BIOS

Rolul BIOS-ul este acela de a acţiona ca o interfaţă între software si hardware, impunând un standard ce trebuie respectat de toţi manufacturierii de calculatoare. O parte din rutinele BIOS, cum ar fi cele ce realizează testele de verificare a resurselor hardware (POST – Power On-Line Self Test) se găsesc în memoria ROM, în timp ce partea de BIOS cuprinsă în fişierul IOSYS.SYS se încarcă în partea inferioară a memorie RAM. Modul de mapare a memorie este standardizat, existând mai multe versiuni de BIOS: Award, Phoenix etc. Dacă în primele versiuni, BIOS-ul oferea întreruperi cu numere cuprinse între 00h şi 1Fh, arhitecturile moderne oferă seturi extinse de întreruperi, tabelul 11.2.

Numărul serviciilor oferite de BIOS este foarte mare, de aceea în tabel sau omis multe dintre acestea. După cum se observă, unele dintre acestea oferă un mod de întrerupere a procesorului de către dispozitivele ataşate la acesta, aşa numitele IRQ-uri (Interrupt Request), prezentate, în mare majoritate, în tabelul 11.1, în timp ce altele oferă o modalitate de accesare a dispozitivelor de către programator prin intermediul instrucţiunii int.

Modul de întrebuinţare a întreruperilor de acces la dispozitive, este comun atât serviciilor BIOS cât şi celor DOS. Pe lângă numărul de întrerupere, care indică dispozitivul ce se vrea accesat, programatorul trebuie să specifice şi funcţia pe care o doreşte realizată. Aceasta se specifică prin registrul ah (câteodată, în funcţie de serviciu şi funcţie, se foloseşte şi AL, BX şi DX), după cum se observă în exemplu următor:

;pozitionare cursor:


;pregatire parametrii


mov
ah, 02h


;functia 02h


mov
bh, 00h


;pagina video


mov
dh, 10


;linia


mov
dl, 12


;coloana


;apel intrerupere


int
10h


;serviciu video


;nu se intoarce nici un rezultat

Acest exemplu poziţionează cursorul pe linia 10, coloana 12, pe prima pagina video, apelând funcţia 02h a serviciului 10h. Modul de folosire al întreruperilor este acelaşi, ceea ce diferă fiind doar modul de transmitere al parametrilor şi de primire a rezultatelor, acolo unde cazul.

Tabel 11.2


Număr întrerupere
Semnificaţie (serviciu)

00h
Divizare la zero

01h
Modul single-step (vezi flag tf)

02h
Nemascabil (NMI)

03h
Breakpoint

04h
Depăsire (overflow) (vezi flaguri of si ov)

05h
Tipărire ecran

06h
Cod de operaţie nevalid

07h
Nu există coprocesor matematic

08h
IRQ0 Timer

09h
IRQ1 Tastatură

0Ah
IRQ2 cascade (trecerea pe controllerul 2 de întreruperi)

0Bh
IRQ3 COM 2/4

. . . . .
. . . . .

10h
Serviciul video

11h
Lista echipamentelor

12h
Mărimea memoriei convenţionale

13h
Disk I/O

14h
Port serial

. . . . .
. . . . .

40h
Vector către funcţiile int 13h referitoare la dischetă

41h
Vectori către structurile referitoare la hard discuri

43h
Tabela de caractere EGA si VGA

46h
Vectori către structurile referitoare la hard discuri

4ah
Alarmă utilizator

67h
Funcţii ale managerului de memorie expandată

70h
IRQ8 Real Time Clock

. . . . .
. . . . .

76h
IRQ14 hard disk

77h
IRQ15 (rezervat)

11.3.2  Întreruperi DOS

Întreruperile oferite de stratul BDOS, oferind în special rutine de acces la disc, fac invariantă interfaţa în raport cu diferitele particularităţi hardware din modelele de PC-uri oferite de fabricanţi. Deşi mai puţine la număr, întreruperile DOS oferă un mare număr de funcţii, tabelul 11.3.

Tabel 11.3

Număr întrerupere
Semnificaţie (serviciu)

20h*
Terminare program

21h
Servicii DOS

22h
Adresa de revenire din program după terminarea acestuia

23h
Adresa de revenire după terminare cu Control-Break

24h
Adresa de revenire după o eroare critică

25h/26h
Acces direct la disc

27h*
Terminare program cu rămânere rezindent în memorie

28h
Procesor neocupat (idle)

29h
Acces rapid la ecran

2Eh
Executare comandă DOS

2Fh
Multiserviciu (DoubleSpace, spooler, control TSR, etc.)

31h
Servicii DPMI (folosite sub modul Windows 386Enh)

33h
Suport mouse

67h
Servicii ale managerului de memorie expandată (HIMEM.SYS)

*Aceste servicii sunt rămase de la versiunile iniţiale de DOS, existând 

servicii DOS mai noi şi mai uşor de folosit care le înlocuiesc.

În continuare se prezintă un exemplu de folosire a serviciului DOS 21h:

mov
ah, 09h
;functia de afisare sir

mov
dx, sir
;in ds:dx se pune adresa sirului

int
21h

;numarul serviciului

Secvenţa anterioară realizează afişarea unui sir terminat cu caracterul $ prin apelul funcţiei 09h a serviciului 21h. Acest cod demonstrează modul similar de folosire al întreruperilor atât pentru stratul BIOS cât şi pentru stratul DOS. Pentru accesarea resurselor hardware în limbaj de asamblare există trei niveluri de programare:

· la nivel DOS

· la nivel BIOS

· la nivel de bază.

Aceste modalităţi au fost aşezate în ordinea crescătoare a complexităţii programării. Astfel, nivelul DOS oferă servicii de un nivel mai înalt decât nivelul BIOS, în timp ce programarea la nivel de bază este cea mai grea folosind instrucţiunile in şi out de lucru cu porturi. Această modalitate de accesare a resurselor necesită cunoştinţe avansate si o documentaţie pe măsură, pe când în programarea cu întreruperi, cunoscând modul general de utilizare al acestora, programatorul trebuie doar să aibă la dispoziţie documentaţia pentru fiecare funcţie referitoare la modul de transmitere al parametrilor şi de preluare a rezultatelor. Cu toate acestea, unele sarcini realizate cu ajutorul întreruperilor necesită cunoştinţe profunde hardware, precum şi folosirea concomitentă cu instrucţiunile in şi out, după cum se va observa în exemplu următor.

11.4 Exemplu de folosire al întreruperilor

Folosirea întreruperilor necesită cunoştinţe despre efectele colaterale ale acestora. Programatorul, în funcţie de întreruperea folosită, trebuie să ştie mecanismele ce se află în spatele serviciului. Pentru a scrie un program care să nu afecteze buna funcţionare a sistemului, el trebuie să ţină cont atât de buna funcţionare a propriului program, luat separat, cât şi de funcţionarea a celorlaltor programe executate în paralel cu acesta, precum şi a dispozitivelor periferice care folosesc întreruperi. Toate aceste probleme pe care le are de înfruntat un programator nu au o forma generală, ele sunt specifice fiecărui tip de întrerupere, neputându-se pune un diagnostic concret unei erori date decât ţinându-se cont de întreruperile folosite.

În continuare, pentru exemplificarea problemelor ce apar la folosirea unei întreruperi, se prezintă un exemplu, devenit clasic, preluat din cartea “Mastering Turbo Assembler” a lui Tom Swan, cunoscut autor de cărţi de programare. Acest exemplu, relativ simplu la prima vedere, îşi propune să folosească întreruperea IRQ 0h, cu numărul 08h (Timer), pentru executarea unui cod la intervale regulate de timp. Acest cod nu face altceva decât să încetinească programul prin adăugarea de pauze mici, fiind folositor în cazul în care se doreşte depanarea unui program care se dovedeşte a fi prea rapid pentru observarea operaţiilor pe care le realizează, depanarea pas-cu-pas fiind o alternativă greoaie de folosit.

În toate calculatoarele compatibile IBM PC, există un timer hardware, care generează, la intervale egale de timp (de 18.2 ori pe secundă), un semnal de întrerupere pe linia 0h către controllerul 8259, producând o întrerupere cu numărul 08h, de maximă prioritate. Serviciul cu acest număr are două sarcini. Prima constă în incrementarea unei variabile pe 32 de biţi, care indică numărul de cicluri ale timerului de la pornirea calculatorului, folositoare pentru diferite funcţii ce indică timpul de rulare. A doua sarcină controlează modul de funcţionare al motorului unităţii de dischetă. Această sarcină este datorată timpului necesar motoraşului pentru a ajunge la turaţia necesară. Dacă acesta ar fi oprit după fiecare citire/scriere, între aceste operaţii există timpi morţi necesari rotorului pentru a ajunge la turaţia optimă, lucru care ar duce la o încetinire mare a operaţiilor cu discheta. Soluţia a constat în menţinerea în turaţie a motoraşului atâta timp cât un cronometru, controlat de întreruperea 08h, nu ajunge la zero.

În scrierea unui program care foloseşte întreruperea, trebuie ţinut cont de aceste sarcini suplimentare. Codul programatorului, care este apelat la intervale regulate de timp, nu trebuie să dureze mult. În cadrul acestuia nu trebuie să se oprească primirea întreruperilor de către procesor (instrucţiunea cli) pentru mai mult de 1/18,2 secunde, pentru a nu se afecta buna funcţionare a unor componente esenţiale ale sistemului. După cum se observă, folosirea unei întreruperi relativ banale, ridică în mod neaşteptat, pentru necunoscători, probleme mari în funcţionarea unor periferice şi a unor programe relativ independente.

Pe lângă aceste doua sarcini, serviciul timer execută o întrerupere software cu numărul 01Ch (User timer interrupt), care dă posibilitatea programatorului de a executa codul său. Prin instalarea propriei rutine de tratare a întreruperii 01Ch, aceasta este apelată de aproximativ 18.2 ori pe secundă. Regularitatea apelării acesteia nu este garantată. După cum s-a afirmat deja, întreruperile nu pot să întrerupă instrucţiunea curentă. Dacă aceasta necesită multe cicluri procesor, executarea întreruperii se amână nedorit de mult, astfel încât premisa executării de 18.2 ori pe secundă a codului să fie neadevărată.

În continuare se prezintă codul programului sursă, urmând ca fiecare grup de instrucţiuni să fie prezentat în amănunt:

;****************************************

%TITLE “Exemplu de folosire a intreruperii 01Ch"

;****************************************

IDEAL

.MODEL   small

.STACK   256

delay

EQU
0010h
; pauza

BIOSData
EQU
040h
; adresa segmentului de 

;date BIOS

LowTimer
EQU
006Ch
; valoarea timerului (partea mai

; putin semnificativa)

PIC8259
EQU
0020h
; adresa de port a 

;cipului 8259 PIC

EOI

EQU
0020h

 valuarea “End of 

;interrupt”

DATASEG

codIesire
DB
0

timerSeg
DW
?

; adresa serviciului initial

timerOfs
DW
?
; Int 1Ch ISR

CODESEG

Start:  

mov ax, @data
;initializare registru

mov ds, ax                  
; de date

mov es, ax                 
; es = ds

mov [word cs:difference],delay  ; stabilire pauza

push es                      
;salvare registru es

mov ax, 351Ch               
;obtinerea vectorului de ; intrerupere 1C

int 21h                     
; apel serviciu DOS

mov [timerSeg], es          ; salvare segment

mov [timerOfs], bx          
; salvare offset

pop es                      
; restaurare es

push ds                      
; salvare registru ds

mov ax, 251Ch               
; setarea serviciului 1C

push cs                      
; ds = cs

pop ds                      


mov dx, offset Incetinire ; adresa de inlocuire

int 21h                     
; setare adresa

pop ds                      
; restaurare ds

@@10:

call KeyWaiting      


jz @@10                    
; asteptare apasare tasta

push ds                      
; salvare ds

mov ax, 251Ch               
; setare vector 1C

mov dx, [timerOfs]          
;adresa de segment initiala

mov ds, [timerSeg]          ;offset-ul initial

int 21h


; restaurare serviciu initial

pop ds                      
; restaurare ds

Exit:

mov ah, 04Ch

mov al,  [codIesire] 

int 21h                 

;****************************************

; Incetinire - Procedura apelata de aprox. 18,2/sec

; ****************************************

inProgress
DB
0       ; flag

difference
DW
0       ; timpul de incetinire

PROC
Incetinire

;testare flag inProgress pentru a vedea daca o copie

;anterioara a procedurii ruleaza

cmp [byte cs:inProgress], 0
; verificare

jne @@99                    

; ruleaza, terminare

; procedura

inc [byte cs:inProgress]    
; nu ruleaza,

; incrementare flag

sti




; pot sa aiba loc alte

; intreruperi

push ax                      
; salvare registrii

; folositi

push ds

push dx

mov al, EOI           

; al = end-of-interrupt

out PIC8259, al             



mov ax, BIOSData    

; adresa de date BIOS

mov ds, ax                 
; folosire ds

mov ax, [word LowTimer]     
; valuare timer

@@10:

mov dx, [word LowTimer]     
; incarcare valoare

; timer in dx

sub dx, ax                  

; scadere valoare 

; noua - veche

cmp dx, [cs:difference]     
; comparare cu 

; pauza

jb @@10                    
; cicleaza pana la 

; depasire

cli                           
; dezactivare intreruperi 

; in timpul restaurarii

dec [byte cs:inProgress]            ; flag pe zero

pop dx 
; restaurare registri

pop ds

pop ax

@@99:

iret



; retur intrerupere

ENDP
Incetinire

PROC    KeyWaiting

push ax          


; salvare ax

mov ah, 1        


; functia de 

; verificare buffer

int 16h          


; serviciu tastatura 

; BIOS

pop ax           


; restaurare ax

ret              


; retur

ENDP    KeyWaiting

END     Start        


; sfarsit program

La rularea programului se instalează procedura Incetinire ca serviciu al întreruperii 1Ch, urmând ca, la apăsarea unei taste, programul să se termine. Listing-ul începe prin definirea unor simboluri. Pe lângă definirea pauzei (delay), se definesc o serie de simboluri, care presupun cunoaşterea modului de mapare al variabilelor BIOS în memoria RAM. Astfel, adresa 40h semnifică începutul zonei de mapare BIOS, pe când 40h::6Ch este locaţia de memorie unde valoarea curentă a timerului se află. Urmează segmentul de date, cu definirea unor variabile ce vor fi folosite pentru reţinerea adresei vechii rutine de tratare a întreruperii 1Ch (timerSeg: timerOfs). 

După liniile de început a segmentului de cod, urmează obţinerea rutinei de tratare a întreruperii 1Ch. Se foloseşte întreruperea DOS 21h (int 21h), funcţia 35 (registrul ah = 35, registrul al = 1Ch). Vectorul obţinut se salvează în timerSeg:timerOfs. Urmează setarea ca serviciu a rutinei Incetinere prin funcţia 25 a întreruperii 21h. După stabilirea noului serviciu, se trece la executarea unei bucle din a cărei repetare se iese prin apăsarea unei taste. Pe parcursul acesteia, deşi aparent nimic nu se întâmplă, procedura Incetinere este apelată în mod repetat de către serviciul întreruperii 08h (timer). La ieşirea din ciclu, se reface vechea rutină de tratare a întreruperii 1Ch. Acest lucru este necesar, deoarece, odată cu terminare programului, adresa, spre care Incetinire puncta, devine invalidă, iar apelul realizat la această adresă de către serviciul timer duce, inevitabil, la o cădere  de sistem.

Se trece acum la rutina de tratare a întreruperii, Incetinire. Deoarece această rutină poate fi apelată în orice moment, între două apeluri de instrucţiuni din programul principal, în timpul unui apel către sistemul de operare, sau în timpul unei alte întreruperi, valoarea segmentelor ES şi DS nu punctează întotdeauna către segmentele de date. Pentru accesarea unor variabile din cadrul segmentelor de date, trebuie să fim siguri că aceste registre identifică în mod corect aceste segmente. O soluţie este executarea unei secvenţe de cod care în mod uzual se găseşte la începutul programului principal:




push
ds


mov
ax, @data




mov
ds, ax




. . . . .

urmată, la sfârşitul întreruperii, de refacerea segmentului ds:




. . . . .




pop
ds




iret

O altă soluţie, mai elegantă şi mai simplă, este declararea variabilelor în cadru segmentului de cod. Acest lucru se realizează în procedura Incetinire prin:



inProgress
DB
0       ; flag

difference
DW
0       ; timpul de incetinire

Pentru accesarea variabilelor se indică în mod explicit folosirea registrului CS, ca în cazul instrucţiunii:

cmp     [byte cs:inProgress], 0

Dacă nu se specifică registrul CS, asamblorul consideră ca registru segment registrul DS, iar adresa de memorie accesată, în cel mai bun caz, dacă este validă, nu este cea dorită.

După cum arătat deja, întreruperea 08h joacă un rol important în funcţionarea unui sistem de calcul. Rutina apelată de timer nu trebuie să dureze mult, iar pe parcursul acesteia celelalte întreruperi nu trebuiesc blocate. Această situaţie face posibilă apariţia apelului rutinei Incetinire chiar în timpul rulării acesteia. Nimic nu împiedică procesorul să primească încă o întrerupere 1Ch din partea timerului în timpul execuţiei procedurii Incetinire, dacă aceasta durează prea mult. Astfel, acestă rutină devine reentrantă. Din păcate scrierea unei funcţii reentrante ridică probleme sulpimentare, chiar rutinele din cadrul BIOS-ului si DOS-ului nefiind reentrante. Posibilitatea reapelării procedurii Incetinire poate să ducă la depăşirea stivei dacă procedura durează prea mult, iar lanţul de reapelării continuă fără oprire. Mai mult, folosirea unor variabile globale face ca execuţia să ia o întorsătură neprevăzută, o singură instanţă a acestora fiind folosită la toate apelurile. După cum se observă folosirea variabilei difference, face imposibilă folosirea rutinei Incetinire în mod reentrant. Soluţia la această problemă (soluţie pe care o folosesc şi unele funcţii BIOS/DOS, cum ar fi Print Screen) este folosirea unui flag (în cazul de mai sus inProgress), care să indice dacă o instanţă a rutinei este deja apelată:

cmp     [byte cs:inProgress], 0

Dacă inProgress este mai mare ca zero, instanţa curentă a apărut ca urmare a blocării altui apel al rutinei, şi se sare la sfârşitul procedurii. Dacă este zero, instanţa curentă este singura apelată în acest moment şi se poate continua execuţia ei. Urmează incrementarea flagului pentru indicarea acestui lucru.

După luarea acestor măsuri, urmează altele: permiterea altor întreruperi (sti) şi salvarea regiştrilor folosiţi în stivă. Deşi, la prima vedere, instrucţiunea sti pare suficientă pentru activarea întreruperilor, datorită faptului că întreruperile timerului sosesc prin intermediul cipului 8259, acest lucru nu este adevărat. Deşi instrucţiunea sti setează flagul if, permiţând indicarea către procesor a semnalelor de întrerupere, primirea de către acesta a întreruperilor necesită trimiterea către portul 8259 (cu valoarea 20h) a unei comenzi end-of-interrupt (EOI):

PIC8259
EQU     0020h ; adresa de port a cipului 8259 PIC

EOI

EQU     0020h ; valuarea “End of interrupt”


. . . . . .



mov     al, EOI      ; al = end-of-interrupt

out       PIC8259, al             




. . . . . .

Urmează citirea valorii curente a timerului şi folosirea acesteia în cadrul unei bucle pană la depăşirea pauzei (variabila difference cu valoarea luată din delay). Deşi la o simplă privire pare că variabila punctată de adresa LowTimer are o valoarea constantă, ea este actualizată automat de către serviciul timer, iar simpla recitire a acesteia preia o valoare actualizată.

Ultima parte a procedurii conţine o restaurare a regiştrilor folosiţi, precum şi o resetare a flagului inProgress, precedate de o mascare necesară a întreruperilor (instrucţiunea cli). O atenţie deosebită trebuie avută în vedere în legătură cu variabilele salvate pe stivă atunci când se scrie o rutină de tratare a unei întreruperi. Singurul registru de segment cu valoare corectă pe tot parcursul execuţiei rutinei este CS, în timp ce ES, DS şi SS au valori incerte, în funcţie de codul în curs de execuţi înainte de întrerupere. În exemplul anterior se salvează trei regiştri în stivă. Dar, în care stivă? În stiva proprie DOS-ului, în stiva programului principal sau în stiva locală altei rutine de tratare a unei întreruperi? Este această stivă suficient de mare? De obicei, se consideră sigură salvarea unei mici cantităţi de date într-o astfel de stivă necunoscută. Referinţe de programare prevăd rezervarea pentru spaţiul de stivă a unei memorii mai mari decât strictul necesar, deci presupunând ca există spaţiu de salvare pentru aceşti regiştrii pare rezonabil. Totuşi, rutine, care necesită salvări mari de date pe stivă, trebuie să apeleze la o stivă locală. De fiecare dată când se schimbă segmentul de stivă, trebuie ţinut cont de particularităţile specifice procesoarelor din familia 8086. Astfel, 8086 dezactivează în mod temporar întreruperile pentru următoarea instrucţiune de fiecare dată când se asociază nouă valoare unui registru segment. Deci, în secvenţa de cod:




mov
ax, offset segmStiva




mov
ss, ax




mov
sp, offset sfarsitStiva

întreruperile nu pot opri execuţia codului între instrucţiunile mov ss, ax şi mov sp, offset sfarsitStiva. Dacă acest lucru nu s-ar realiza în mod automat, iar o întrerupere ar avea loc chiar între cele doua instrucţiuni, codul de tratare apelat va folosi vechea valoare a deplasamentului împreună cu noua valoare a segmentului de stiva, o situaţie cu urmări catastrofale. Prin urmare, pentru folosirea acestei metode de protecţie, orice modificare a unui segment, trebuie urmată de actualizarea deplasamentului.

Pentru o rutină de tratare a unei întreruperi, declararea unui segment de stivă locală de 512 baiţi face astfel:




ALIGN




stivaLocala
DB
512 dub (0) 




sfarsitStiva
=
$

unde directiva ALIGN este necesară pentru alinierea stivei la un multiplu de cuvânt procesor. Se declară, apoi, o serie de variabile pentru reţinerea vechilor regiştri:




ssVechi
DW
0




spVechi
DW
0




axVechi
DW
0

În cadrul rutinei se salvează vechile valori ale regiştrilor şi se stabileşte noua stivă:

mov

[cs:ssVechi], ss

mov

[cs:spVechi], ss

mov

[cs:axVechi], ax

mov

ax, cs

mov

ss, ax

mov

sp, offset sfarsitStiva

Urmând ca la terminarea rutinei să se revină la vechile valori:


mov
ss, [cs:ssVechi]
;aceasta este ordinea de setare


mov
sp, [cs:spVechi]
;a registrilor segment:offset !!!


mov
ax, [cs:axVechi]

După cum se observă, o sarcină aparent simplă, cum este cea de folosire a timer-ului, necesită cunoaşterea profundă a implicaţiilor pe care le implică utilizarea întreruperilor, nerespectarea regulilor mai sus amintite ducând inevitabil la terminarea anormală a programului sau, chiar, la o cădere de sistem.

11. 5 Concluzii

Folosirea întreruperilor nu este o sarcină uşoară. De multe ori soluţia cea mai simplă este folosirea unui limbaj de nivel înalt. Dar, de şi mai multe ori, ceea ce oferă aceste limbaje, referitor la folosirea directă a hardware-ului, nu este de ajuns. Prin urmare, folosirea limbajului de asamblare şi a întreruperilor devine inevitabilă. Sistemele de operare noi (Windows NT, Windows 2000) nu permit accesarea directă de hardware, oferind soluţii de nivel înalt pentru accesarea acestuia. Totuşi, scrierea de programe specifice şi dedicate unor sarcini precis determinate, cum este scrierea unor drivere, accesarea unor echipamente noi, necesită realizarea unui cod care să ruleze în mod nucleu şi care să acceseze resursele direct. Astfel, programarea cu întreruperi îşi păstrează actualitatea.

































































